
Jason & Glenwick, eds. 

Chapter 10/Web Appendix guide 
 

Web Appendix Guide for Chapter 10/Multilevel Modeling  
 

The Web Appendix materials for Chapter 10 include data sets, SAS syntax, and this set of 
instructions for loading the data sets into SAS and producing SAS output files for Multilevel 
Logistic Modeling (MLM). These materials provide the information necessary to run the 
analyses presented in the chapter.     
 

Overview Materials 
 Basic SAS Instructions.  This document describes basic SAS operations such as:  

o Loading data into SAS 
o Basic SAS syntax and tips 

 Variable_Description. This document describes the variables and gives specific items. 
 Multilevel Logistic Modeling. This document describes in further detail the use of 

multilevel modeling for logistic outcomes. 
 Computing the Intraclass Correlation in SAS. This document describes how to 

compute the intraclass correlation. 
 Setting up the Data File. This document describes how to set up the data set for analysis 

in SAS. 
SAS Syntax: All SAS code is given in a SAS syntax file and in a Word file. 

 Creating Multilevel Datasets.   
o Use this syntax to create a multilevel dataset from the “MLM_build” person-level 

dataset.   
o Use this syntax to create centered variables, group Level II variables, and the 

“MLM_group” data set.  Running the entire syntax will result in the 
“MLM_group” and “MLM_complete” datasets.  

 MLM Graphing.  Use this syntax to generate descriptive graphs. 
 MLM Continuous Prioritization.  Use this syntax to model the continuous prioritization 

outcome. 
 MLM Logistic Participation.  Use this syntax to model the discrete outcome of 

participation. 

DATA SETS AND ONLINE TABLES AND FIGURES IN THIS WEB APPENDIX 
 
Data sets: All datasets are given in SAS and SPSS file formats. 

 MLM_build.  A person-level data set containing only individual level variables.  
The user can practice group- and grand-mean centering individual level variables, 
and can create group Level II variables. 

 MLM_group.  A group-level data set for computing group-level descriptive 
statistics. 

 MLM_complete.  The complete data set ready for analysis. 
 
Online Tables and Figures (separate links in the Web Appendix folder):  

 Online Table 10.1: Example Data File Structure 
 Online Table 10.2: Example Level 1 and Level 2 Correlation Matrix 
 Online Table 10.3: Example Table to Report MLM Results 
 Online Figure 10.1: Regression Lines for Bridging and Social Justice Prioritization 

for Each Congregation 
 Online Figure 10.2: Example Follow-Up Interaction Plot for SJ Prioritization  
 Online Figure 10.3: Example Follow-Up Interaction Plot for SJ Participation 

BASIC SAS INSTRUCTIONS 
 
SAS is a flexible and powerful statistical program that uses syntax to manipulate variables and to 
conduct analyses.  The following instructions are a short tutorial to give the reader basic facility 
in SAS, including the loading of data sets, the loading of data into the workspace, working with 
variables, and loading SPSS files into SAS.  Though not complete, these SAS tips may be 
helpful when practicing multilevel modeling. 
 
Creating a SAS Data Set from SPSS 

1. In SPSS, save the data file as type SASv7 + Windows long extension (*.sas7bdat). 
 
Loading Data Into SAS 

1. Save your SAS data set to a folder on your hard drive that is dedicated to SAS datasets.  
For this exercise, you can label your SAS data set “testaa”. 

2. Create a folder on the desktop called “SAS”.  Every computer will be different, though 
you can trace the path information by looking at the top of the folder name.  Move 
“testaa” into this folder. 

3. In the SAS program editor, use the following code to designate the path to find the files 
(note, this is an example, your path will be different): 

 
libname sasdata "C:\Users\Nathan\Desktop\SAS"; 
run; 

 
data testbb; 
set sasdata.testaa; 
run; 
 
The first line of the “libname” command creates a new library called “sasdata”, and the 
“C:\” tells SAS where to find the files to put in the library. 

 
The next “data” line of code tells the program to create a new file titled “testbb” from the 
data set “testaa”, located in the library sasdata. This uses the file “testaa” from the library 
to create a new file called “testbb” in your work folder, called the workspace. 

 
You can also save data sets from your work folder to the SAS library, which then saves 
the data set on your computer.  This is not telling SAS to use the file “testbb” from your 
work folder, and to create a new file in the library called “testdd”: 
 
data sasdata.testdd; 
set testbb; 
run; 

 
Once you have loaded your data set, check the work folder to make sure it is there!  
You are now ready to work with this data set. 

Basic SAS syntax and tips 
 Lines MUST end with a semicolon (;). 
 Comments can be typed between asterisks (*); note how these turn the lines green. 
 Variable names should be 1 to 32 characters in length, beginning with a letter. 
 Check the log file to make sure your program runs okay. 
 Save your syntax for everything! 
 Data steps  

o Create and modify data sets. 
 Proc steps = Procedures 

o Performs some procedure or analysis with the data set. 
 
Manually Entering a Data Set Into SAS  

 
Sample SAS program: 

*Reads data that you enter into a SAS data set, stored in WORK folder. 
*Missing data are represented by a single period (.). 
*variables are separated by spaces. 
*data are separated by spaces. 
*note that the $ after gender designates that the variable is designated by a character. 

 
data lab1; 
input id gender $ var1 var2 var3; 
cards; 
1 F 79 80 92 
2 M 69 87 79 
3 F 78 76 90 
4 F 73 85 88 
5 M 80 86 94 
; 
run; 
 

To run a program: 
1. Select and copy your code. 
2. Click on the little person icon near the top of your screen to submit or run your program.  

 
 

Basic SAS Commands 
 
Sorting Data: 
 

proc sort data=testaa; 
by gender; 
run; 

Look at data set and observe order to see if sorted by gender. Now sort by id: 
 

proc sort data=lab1; 
by id; 
run; 

 
Descriptive Statistics: 
 

proc means data=lab1; 
var var1 var2 var3; 
run; 

 
Modifying Variables, Merging Data Sets:  
 

To rename variables: 
 

data testff; 
set testaa; 
act = var1; 
gpa =  var2; 
run; 

  
This code takes the original variables name, “var1,” and renames it as “act”. It takes the 
values from the original data set “testaa” and creates a new data set “testff”, where the 
variables are renamed. 

 
To delete variables: 

 
data testgg; 
set testaa; 
drop var1 var2; 
run; 

 
This code takes the original variables, “var1” and “var2” from the data set “testaa” and 
drops them in the new data set “testgg”. 

 
To re-code variables from character to numeric: 

 
data testhh; 
set testaa; 
if gender = 'M' then gender1 = '1'; 
if gender = 'F' then gender1 = '2'; 
gendernum = input(gender1,1.0); 
*if you want to dummy code with male as reference group; 
if gendernum = 2 then genddum = 1; 
if gendernum = 1 then genddum = 0; 
run; 

Linear transformations: 
 

data testii; 
set testaa; 
var1a = var1 + 25; 
var2a = var2*15; 
varscale = (var1 + var2 + var3)/3; 
run; 

 
To create new variables “var1a” and var2a”, SAS takes the original variables var1 and 
var2 and computes the mathematical operation of addition (for var1) or multiplication 
(var2).  Any mathematical operation can be used. Varscale takes var1, var2, and var3 to 
form a scale based on the mean of these variables. 

 
Creating groups: 

 
data testjj; 
set testaa; 
if var1 < 75 then var1grp = 0; 
if var1 = 75 then var1grp = 0; 
if var1 > 75 then var1grp = 1; 
run; 

 
Similar to linear transformations, new variables can be created by forming groups.  This 
is helpful when computing groups based on median splits. 

 
Adding a variable and merging data sets: 

 
data test1; 
input id var1 var2 var3; 
cards; 
1 3 2 4 
2 4 2 1 
3 5 2 3 
4 1 3 2 
5 4 2 3 
; 
run; 

data testextra; 
input id var4; 
cards; 
1 99 
2 76 
3 68 
4 38 
5 99 
; 
run; 

 
To add a variable, you need to merge two data sets together and match by an identifier: 
 
First, sort each data set by the identifier, id:  

 
proc sort data=test1; 
by id; 
run; 
 
proc sort data=testextra; 
by id; 
run; 

 

Now, create a new data set that merges these together: 
 

data testcombined; 
merge test1 testextra; 
by id; 
run; 

 
The data set “testcombined” should contain variables var1, var2, var3, and now the added 
variable var4. 

VARIABLE DESCRIPTION 
 

These data are individual responses to the Church and Community Planning Inventory (CCPI), collected by Dudley (1991) as part of project 
conducted by the Center for Church and Community Ministries. The project was funded by the Lilly Endowment, and the data are available online 
at http://www.thearda.com. The final data set has 5,123 individuals nested within 62 Christian religious congregations. 

 
Outcome Variables Description Scale Anchors 
     Social Justice Prioritization Four item scale (α = .82) assessing the priority members place 

on congregational social justice actives such as efforts to 
relieve poverty and hunger, promote world people and human 
rights. 

1 (lowest) to 5 (highest) 

     Social Justice Participation One item asking if people gave time (or not) to congregational 
social justice programs. 

1 (any amount of time) and 0 (no time) 

 
Predictor Variables   
     Bridging Social Capital One item assessing agreement that the congregation values 

cooperative projects with churches of other denominations. 
1 (disagree strongly) to  
4 (agree strongly) 

     Frequency of Religious Attendance Three item scale (α = .64) assessing frequency of attendance at 
worship services, educational courses, and other church 
activities. 

1 (never) to 7 (every day) 

     Theological Conservativism One item that assessing attitudes toward Scripture as literal or 
metaphoric. 

1 (literal) to 4 (metaphoric) 

Hypotheses          Corresponding Model from Table XX 
     Within  There will be a positive association between individual bridging and social justice*. B and D 
     Between  There will be a positive association between congregational bridging and social 

justice. 
B 

     Contextual Congregational bridging will predict social justice over and above individual 
bridging. 

D 

     Cross-Level The association between frequency of religious participation and social justice will be 
moderated by theological orientation. It is predicted that within liberal congregations 
the association will be positive and stronger than in conservative congregations. 

C 

 Note. *social justice refers to both SJ prioritization and SJ participation.

Social Justice Prioritization Four Item Scale: α = .82 
There are many tasks that a church can do. Of those listed below, what priority would you give to each 
for your church?  
1. Encourage individual members to support local social reforms to relieve poverty and hunger. 
2. Provide organizational support to groups attempting to promote world peace and justice. 
3. Provide opportunities for channeling members into actively promoting human rights and well-

being. 
4. Develop church programs which would help people understand local programs and issues. 

Response options ranged from 1 (none) to 5 (highest) 
 
Social Justice Participation Item: 
     Those who answered yes to the question, “Does your congregation participate 
     directly in any programs to provide a community social service or promote peace and 
     justice?” were then asked,  

1. How much time have you been able to give to these programs? 
Response options ranged from 1 (none) to 5 (More than five hours a week) 

 
Theological Orientation Item:   

Which of the following statements best expresses your view of the Bible?” 
1. The Bible is the actual Word of God and is to be taken literally.  
2. The Bible is the inspired Word of God and its basic moral and religious teachings are clear and 

true, even if it does contain some human error.  
3. The Bible is a record of many different people’s response to God and because of this, people and 

churches today are forced to interpret for themselves the Bible’s moral and religious teachings.  
4. The Bible is a valuable book because it was written by wise and good people, but I do not believe 

it is really God’s word.  
 
Frequency of Religious Participation Three Item Scale: α = .64 
 How often do you personally do the following: 

1. Attend worship service? 
2. Participate in religious courses or Sunday school? 
3. Participate in other church activities other than worship?     
Response options ranged from 0 (never) to 7 (every day) 

 
Social Capital Bonding Item: 
 Of your five closest friends, how many are members of this congregation?  
 Response options ranged from 0 (0 friends) to 5 (5 friends) 
 
Social Capital Bridging Item: 

To what extent do you agree that each statement describes your congregation?  A “Don’t Know” 
response is provided, but please use it only when absolutely necessary. 
1. Cooperative projects and joint workshops with churches of other denominations are highly valued.  

 Response options ranged from 1 (don’t know) through 5 (agree strongly) scale. 

MULTILEVEL LOGISTIC MODELING 
 
 Multilevel logistic modeling can be used to test the same within, between, contextual, and 
cross-level interactive hypotheses presented in the chapter, except now the dependent variable is 
a logistic outcome such as a “success” or “failure” (Raudenbush & Bryk, 2002; Snijders & 
Bosker, 1999). Community researchers may be interested in logistic outcomes such as recidivism 
(or not), policy change (or not), if people vote yes (or no), and employ MLM to examine how 
settings may influence such outcomes. The same models presented in Table 10.1 of the book 
chapter can be used to test these hypotheses while taking into account possible dependence in the 
data from people being nested in settings.  

The implementation of multilevel logistic regression is very similar to multilevel 
regression, where the same models as those in Table 10.1 of the book chapter are employed to 
predict a logistic outcome. A main difference lies in the interpretation of the parameter estimates 

(i.e., 


 ) given that logistic regression is predicting the probability of “success” (e.g., where 0 if 
“failure” and 1 is “success”). For continuous outcomes, the parameter estimate is interpreted as 
the amount of predicted change in ijY  for every unit increase in ijX . In contrast, the parameter 

estimates (i.e., 


 ) in logistic regression are interpreted using odds ratios.  

To illustrate, consider a parameter estimate of 


  = .21 for the continuous predictor of 
political conservatism predicting if a person votes “yes” or “no.” This estimate indicates that for 
a one unit increase of conservatism, the odds of a “yes” vote are 1.23 times larger. There is a 

direct conversion between the parameter estimate .21 and the odds of 1.23 by taking 
^
e , in this 

case e.21 = 1.23. Thus, as conservatism increases, the odds of a “yes” vote also increase. Most 
statistical programs will give the parameter estimate, a statistical test to determine significance, 
and the odds ratio with a 95% confidence interval. If the 95% confidence interval for the odds 
ratio does not include one, the parameter estimate is significant. For example, if the 95% 
confidence interval for the odds ratio of 1.23 was (1.03, 1.43), this indicates a significant odds 
ratio. A negative parameter estimate (e.g., -.30) indicates a lower probability of success, with the 
associated odds ratio between 0 and 1, in this case 0.74. Thus, multilevel logistic regression uses 
independent variables, at both level I and II, to predict the likelihood of a logistic outcome.  
 Though beyond the scope of Chapter 10, a few issues are presented for the reader 
interested in using multilevel logistic regression. First, there are two major statistical approaches 
to modeling nested dichotomous outcomes: generalized linear mixed models (GLMM, Anderson, 
Verkuilen, & Johnson, in preparation; Hedeker, 2005) and generalized estimating equations 
(GEE, Hedeker & Gibbons, 2006). The main difference between these approaches is that GEE 
treat dependence as a nuisance and tries to remove the dependence by examining “population-
averaged effects” or “marginal models” whereas GLMM tries to model and understand the 
nature of the dependence by examining “subject-specific effects” or “conditional models” (Hu, 
Goldberg, Hedeker, Flay, & Pentz, 1998). There is not agreement as to which approach is best; 
however, when the purpose is to understand within and between types of effects when 
individuals are nested in settings, GLMM for individuals nested in groups may be preferred. 
When the design is longitudinal with strong within-subject dependence, GEE may be more 

appropriate (Hedeker & Gibbons, 2006; Molenberghs & Verbeke, 2005). For this chapter and 
example, GLMM are used since the hypotheses involve within, between, and contextual effects. 
 Moreover, many statistical programs will be able to implement GEE or GLMM. In SAS, 
PROC GENMOD implements GEE whereas PROC GLIMMIX and PROC NLMIXED 
implement GLMM. Statistics is an evolving field, and likewise the statistical programs used to 
implement statistical procedures also evolve. For example, both PROC GLIMMIX and PROC 
NLMIXED implement GLMM, where NLMIXED was an initial procedure now gradually being 
replaced by GLIMMIX (GLIMMIX is available in SAS 9.2, or can be downloaded from 
www.sas.com and used with earlier versions of SAS). The differences between these two 
procedures are beyond the scope of this chapter, though it bears mentioning that NLMIXED is 
not as user friendly, is slower, and fits a narrower range of models than GLIMMIX. However, 
the estimates produced by NLMIXED are not biased whereas the estimates from GLIMMIX are 
slightly downward biased and thus more conservative. In this chapter, we use GLIMMIX 
because it is more user-friendly, and are of the opinion that with continued development 
GLIMMIX will produce estimation methods that are less biased, as is already evidenced by the 
ability of GLIMMIX to use numerical integration methods (Snijders & Bosker, 1999). These 
issues are important to consider when using multilevel logistic regression.   
 
Application of Multilevel Logistic Modeling 

This example focuses on implementing analyses and interpreting results. The same 
models presented in Table 10.1 of the book chapter, and used in the earlier example of a 
continuous outcome, are now used with the dichotomous dependent variable of social justice 
participation (SJ participation is whether people gave time or not to congregational social justice 
activities). The same models are run, and statistical information can be presented as shown in 
Online Table 10.3 in this Web Appendix.. However, the interpretation of the parameters is now 
in terms of odds-ratios, where a larger positive parameter estimate indicates the odds of success 
becoming more likely with a unit increase in the independent variable. For our example, we 
found a positive and significant parameter estimate for religious participation in predicting SJ 

participation (e.g., 


  = .64). This indicates that the odds are 1.90 times greater for SJ 
participation for every unit increase in frequency of religious participation.  

Similar to the process of analysis for continuous outcomes, follow up tests are conducted 
for significant cross-level interactions. In our example there is a significant cross-level 
interaction between frequency of religious participation and congregational theological 
orientation in predicting SJ participation. Follow up logistic regression between frequency of 
religious participation and SJ participation would be conducted separately for liberal and 
conservative congregations (as defined in the book chapter example). For these data, logistic 
regressions show a significant association between frequency of religious participation and social 

justice participation for people in conservative congregations (


 = .53, SE = .05, p < .05, OR = 
1.70, n = 1,608). There is also a significant and slightly stronger association for people in liberal 

churches (


  = .70, SE = .04, p < .05, OR = 2.02, n = 2,227).  
This shows that congregational theological orientation moderates the association between 

frequency of religious participation such that there were increased odds of social justice 
participation with increased frequency of participation, with a more pronounced effect in liberal 
congregations. As displayed in Online Figure 10.3, the logistic curves are plotted for a picture of 

the moderation. The intersection of these two logistic curves indicates that an interaction is 
present. Even though both effects are significant, this graph shows a steeper logistic curve for 
those in liberal congregations, indicating that the effect is stronger. For a full analysis using the 
logistic dependent variable of SJ participation, see Todd and Allen (2010). 

COMPUTING THE INTRACLASS CORRELATION IN SAS 
 

If the statistical program does not automatically give this statistic, compute it by, (1) 
running the “null” model, which includes a random intercept and no predictor variables, (2) 
examine the variance at individual (labeled “residual” or “ 2 ” ) and group (labeled “UN(1, 1)” 
or “ 2 ”) levels, and (3) estimate the intraclass correlation by calculating the ratio of group level 

unexplained variance to the total unexplained variance: 22

2





I . For these data, the 

intraclass correlation for the outcome of social justice prioritization is 095.54.06.
06. I . 

This indicates that 9.5% of the variance on the outcome is explained by the group. Taken in 
conjunction with the descriptive plots, it appears that a random intercept is appropriate to be 
included in the model because congregations have different average SJ prioritization scores. 
Thus, since there is unexplained variance at the congregational level, MLM should be used. 
 

SETTING UP THE DATA FILE 
 

The data file needs to be constructed in a particular manner for multilevel modeling, 
though specifics vary depending on the particular MLM software used. The following describes 
the set up when using SAS, as displayed in Online Table 10.1 in this Web Appendix. First, each 
level I unit (i.e., individual) should have one row in the data set. Second, there should be a 
variable that connects each individual to their group (i.e., a column), in this case membership to 
religious congregation. Third, the outcome variables need to be measured at the level of the 
individual, and should each have their own column. Fourth, group level variables can be created 
by taking the average of the variable for each congregation, with the average then assigned to 
each individual in the group.  

It is helpful to name variables consistently to designate them as level I or level II. For this 
example, “grpMvariable” denotes that the variable is a level II variable formed by the group 
mean. For example, the group mean for bridging in congregation 100 was 3.02. Thus, each 
member of congregation 100 is given the same group mean for grpMbridge. Finally, group- and 
grand-mean centered variables should be created for every participant. Group-mean centered 
variables are labeled “groupCvariable” and grand-mean centered variables labeled 
“grandCvariable.”  
 



15 

 

SAS SYNTAX: CREATING MULTILEVEL DATASETS 
 
OVERVIEW 
The following syntax starts with a person-level data-set, and builds both the group 
("mlm_group") and multilevel ("mlm_complete") datasets. These datasets are provided in the 
data sets folder, and if you just want to practice mlm analysis without building the datasets, you 
can open the other syntax documents for graphing ("mlm_graphing"), modeling SJ prioritization 
("mlm_continuous_prioriziation"), or modeling SJ participation ("mlm_logistic_participation"). 
 
See the "Basic_SAS" section in this Guide for instructions on loading data sets. 
 
libname sasdata "C:\INSERT YOUR PATH TO THE FOLDER WHERE THE DATA FILES ARE 
HERE"; 
run; 
 
data mlm_build; 
set sasdata.mlm_build; 
run; 
 
 
*VARIABLE NAMES USED THROUGHOUT THE ANALYSIS AND SAS SYNTAX 
 
LEVEL I: 
theolbib = theological orientation 
relfreq = frequency of religious attendance 
bond = bonding social capital 
bridge = bridging social capital 
socprior = Social Justice Prioritization 
socpartbin = Social Justice Participation (0 = no participation, 1 = 
participation) 
 
bridgegrand = Grand Mean centered bridging social capital 
bridgegroup = Group Mean centered bridging social capital 
 
bondgrand = Grand Mean centered bonding social capital 
bondgroup = Group Mean centered bonding social capital 
 
relfreqgrand = Grand Mean centered frequency of religious attendance 
relfreqgroup = Group Mean centered frequency of religious attendance 
 
theolbibgrand = Grand Mean centered theological orientation 
theolbibgroup = Group Mean centered theological orientation 
 
Demographics: 
gender = gender where Male = 1, Female = 2 
race = race where White = 1, Black = 2 
age = age in years 
educ = education where larger numbers indicate higher education 
income = income, with higher numbers indicating larger income. 

LEVEL II: 
grpMtheolbib = Group Mean theological orientation for each congregation 
grpMrelfreq = Group Mean frequency of religious attendance for each 
congregation 
grpMbond = Group mean bonding social capital for each congregation 
grpMbridge = Group mean bridging social capital for each congregation 
grpMsocprior = Group mean social justice prioritization;  
 
 
 
FORMING LEVEL TWO VARIABLES: AGGREGATION 
This section calculates the group mean for the church on the variable, creating a new data set 
with the value. The second step merges this data set with the original data set to give each person 
the mean for the group in a new variable. This is done for each level 1 variable, to form the level:   
 
**religious frequency; 
proc means data=mlm_build noprint; 
class church; 
var relfreq; 
output out=relfreq mean = grpMrelfreq; 
run; 
 
 

*proc sort is used before the merge, as SAS requires both data sets to be sorted on the variable 
that is to serve as the matching variable during the merge: 
 
proc sort data=mlm_build; 
by church; 
run; 
 
Deleting the Type=0 is done because this value represents the grand-mean: 
 
data mlm_builda; 
merge relfreq mlm_build; 
by church; 
if _TYPE_=0 then delete; 
run; 
 
Repeat this process for each of the individual level variables that are to be made Level II 
variables, using the data set that was created in the last step: 
 
**theolbib; 
proc means data=mlm_builda noprint; 
class church; 
var theolbib; 
output out=theolbib mean = grpMtheolbib; 
run; 
 
proc sort data=mlm_builda; 
by church; 
run; 
 
data mlm_buildb; 
merge theolbib mlm_builda; 
by church; 

if _TYPE_=0 then delete; 
run; 
 
 
**bond; 
proc means data=mlm_buildb noprint; 
class church; 
var bond; 
output out=bond mean = grpMbond; 
run; 
 
proc sort data=mlm_buildb; 
by church; 
run; 
 
data mlm_buildc; 
merge bond mlm_buildb; 
by church; 
if _TYPE_=0 then delete; 
run; 
 
**bridge; 
proc means data=mlm_buildc noprint; 
class church; 
var bridge; 
output out=bridge mean = grpMbridge; 
run; 
 
proc sort data=mlm_buildc; 
by church; 
run; 
 
data mlm_buildd; 
merge bridge mlm_buildc; 
by church; 
if _TYPE_=0 then delete; 
run; 
 
**social justice prioritization; 
proc means data=mlm_buildd noprint; 
class church; 
var socprior; 
output out=socprior mean = grpMsocprior; 
run; 
 
proc sort data=mlm_buildd; 
by church; 
run; 
 
data mlm_builde; 
merge socprior mlm_buildd; 
by church; 
if _TYPE_=0 then delete; 
run; 

FORMING THE GROUP LEVEL DATA SET 
To form a group level data set with the level two variables, merge the bond, bridge, relfreq, 
theolbib, socprior datasets together.  This data set can be used for group level descriptives: 
 
data mlm_group; 
merge bond bridge relfreq theolbib socprior; 
by church; 
if _TYPE_=0 then delete; 
run; 
 
proc means data=mlm_group; 
var grpMbond grpMbridge grpMrelfreq grpMtheolbib grpMsocprior; 
run; 
 
proc corr data=mlm_group; 
var grpMbond grpMbridge grpMrelfreq grpMtheolbib grpMsocprior; 
run; 
 
data sasdata.mlm_group; 
set mlm_group; 
run;  
 
 
FORMING THE CENTERED VARIABLES 
Variables with “grand” in the title are grand mean centered. Variables with “group” in the title 
are group mean centered. These means are for the entire data set, and will be used for the grand 
mean centered variables: 
 
proc means data=mlm_builde; 
var bridge relfreq theolbib bond; 
run; 
 
The new, centered variables are formed below: 
 
data mlm_buildf; 
set mlm_builde; 
 
bridgegrand = (bridge - 3.0721695); 
bridgegroup = (bridge - grpMbridge); 
 
bondgrand = (bond - 2.0527035); 
bondgroup = (bond - grpMbond); 
 
relfreqgrand = (relfreq - 4.1686418); 
relfreqgroup = (relfreq - grpMrelfreq); 
 
theolbibgrand = (theolbib - 2.1900741); 
theolbibgroup = (theolbib - grpMtheolbib); 
run; 
 
 
CREATING VARIABLE FOR LIBERAL AND CONSERVATIVE CONGREGATIONS 
 
proc freq data=mlm_group; 
table grpMtheolbib; 

run; 
 
After examining the frequencies, if we form conservative congregations as less than 2.19, 
approximately 50% of congregations will be classified as liberal, and approximately 50% as 
conservative: 
 
data mlm_complete; 
set mlm_buildf; 
if grpMtheolbib < 2.141 then grptheolbiblibcon = 0; 
if grpMtheolbib > 2.141 then grptheolbiblibcon = 1; 
run; 
 
The data set "mlm_complete" now contains all individual and group level variables needed for 
analysis. To save this dataset to your computer, use the following: 
 
data sasdata.mlm_complete; 
set mlm_complete; 
run; 

SAS SYNTAX: MLM GRAPHING 
 
OVERVIEW 
This syntax is for producing graphs.  Graphs are very helpful when understanding whether or not 
to include a random intercept or random slope. See the "Basic_SAS" section of this Guide for 
instructions on loading data sets. 
 
libname sasdata "C:\INSERT YOUR PATH TO THE FOLDER WHERE THE DATA FILES ARE 
HERE"; 
run; 
 
data mlm_complete; 
set sasdata.mlm_complete; 
run; 
 
 
SEPARATE REGRESSION LINES FOR EACH CONGREGATION, ALL ON ONE PLOT 
Note that before creating new graphs it is helpful to run the "goptions reset" to reset the graphing 
space; that is why this command is given before any graph: 
 
goptions reset=(axis, legend, pattern, symbol, title, footnote) norotate 
         hpos=0 vpos=0 htext=2 ftext=swiss ctext=black target= gaccess= 
gsfmode=  
         colors= csymbol=; 
goptions device=WIN  ctext=black graphrc interpol=rl; 
 
axis1  color=black   width=2.2  label=('Bridging' );  
axis2  color=black   width=2.2  label=(angle=90 'Social Justice 
Prioritization' );  
 
proc gplot data = mlm_complete; 
plot socprior * bridge = church / haxis=axis1 vaxis=axis2 frame nolegend; 
  title1 'Bridging and Social Justice Prioritization'; 
run; 
 
If you want a legend, delete the "nolegend" command: 
 
proc gplot data = mlm_complete; 
plot socprior * bridge = church / haxis=axis1 vaxis=axis2 frame; 
  title1 'Bridging and Social Justice Prioritization'; 
run; 
 
To examine each church separately: 
 
goptions reset=(axis, legend, pattern, symbol, title, footnote) norotate 
         hpos=0 vpos=0 htext=2.5 ftext=swiss ctext=black target= gaccess= 
gsfmode=  
         colors= csymbol=; 
goptions device=WIN  ctext=black  graphrc interpol=join; 
symbol1 c=black   i=rl   v=none   ; 
symbol2 c=blue  height=3  i=none   v=dot   ; 
axis1  color=black  width=2.5  label=('Bridging') ;  

axis2  color=black  width=2.5  label=(height=2 angle=90 'Social Justice 
Prioritization' );  
goptions nodisplay; 
 
FOR CONSERVATIVE AND LIBERAL CHURCHES: 
 
proc sort data=mlm_complete; 
by grptheolbiblibcon; 
run; 
 
goptions reset=(axis, legend, pattern, symbol, title, footnote) norotate 
         hpos=0 vpos=0 htext=2 ftext=swiss ctext=black target= gaccess= 
gsfmode=  
         colors= csymbol=; 
goptions device=WIN  ctext=black graphrc interpol=rl; 
 
axis1  color=black   width=2.2  label=('Bridging' );  
axis2  color=black   width=2.2  label=(angle=90 'Social Justice Engagement' 
);  
 
proc gplot data = mlm_complete; 
plot socprior * relfreq = grptheolbiblibcon / haxis=axis1 vaxis=axis2 frame; 
  title1 'Bridging and Social Justice Engagement in Liberal and Conservative 
Congregations'; 
run; 
 
To find the graphs, go to the “work” folder and click on “bridge_socprior”.  Each graph is 
labeled with the church number at the top: 
 
proc sort data=mlm_complete; 
by church; 
run; 
 
proc gplot data=mlm_complete gout=bridge_socprior; 
 by church; 
 plot socprior*bridge=1 socprior*bridge=2/  overlay haxis=axis1  vaxis=axis2  
frame  nolegend; 
*  title1 'Religious Frequency x Social Prioritization'; 
run; 
 
These same graphs can be constructed for any variable! 

SAS SYNTAX: MLM CONTINUOUS PRIORITIZATION 
 
OVERVIEW 
This syntax is for modeling the continuous outcome of SJ prioritization. See the "Basic_SAS" 
section of this Guide for instructions on loading data sets. 
 
Put the path to your data set in the “libname” statement: 
 
libname sasdata "C:\INSERT YOUR PATH TO THE FOLDER WHERE THE DATA FILES ARE 
HERE"; 
run; 
 
data mlm_complete; 
set sasdata.mlm_complete; 
run; 
 
 
GENERAL EXPLANATION OF THE SAS CODE 
 
 *Example SAS Code for Continuous Outcomes: 
 *proc mixed data=example1 noclprint covtest method=ml ic; 
 *class church; 
 *model sjprior = GroupCBridge MBridge  / solution ; 
 *random intercept / subject = church type=un; 
 *run; 
 
The model statement in SAS is a regression equation, where the outcome (i.e., SJ prioritization 
“sjprior”) is predicted from group centered bridging and average church bridging. The random 
statement tells SAS that the intercept is random for each of the different churches. The important 
point here is that the terms from the linear mixed model are directly entered into SAS to test the 
specific hypothesis. If one wanted to test the contextual effect, then the terms listed in Table 
10.1/ Model D with the grand-mean centered variables would be entered into the SAS program.  
 
Null model 
The null model is used to calculate the intraclass correlation.  The model is called "null" because 
there are no predictors in the model. Thus the individual and group level variance terms are the 
unexplained variance at these different levels and can be used to construct the intraclass 
correlation: 
 
proc mixed data=mlm_complete noclprint covtest method=ml ic; 
class church; 
model socprior =  / solution ; 
random intercept / subject = church type=un; 
title 'null model'; 
run; 
 
MODEL I: Group mean centered 
Group mean centered is used to tease apart within and between effects, where grpMbridge 
represents the between effect: 

proc mixed data=mlm_complete noclprint covtest method=ml ic; 
class church; 
model socprior = bridgegroup grpMbridge / solution ; 
random intercept / subject = church type=un; 
title 'Model I: group-mean centered model'; 
run; 
 
MODEL II: Grand mean centered 
Grand mean centering is used to test the contextual effect of congregational bridging, or how 
much individual level social justice prioritization can be predicted by congregational bridging 
over and above individual level bridging: 
 
proc mixed data=mlm_complete noclprint covtest method=ml ic; 
class church; 
model socprior = bridgegrand grpMbridge / solution ; 
random intercept / subject = church type=un; 
title 'Model II: grand-mean centered model'; 
run; 
 
Model III: Cross level interaction 
The cross level interaction tests if congregational theological orientation moderates the 
association between frequency of religious participation and social justice prioritization.  Note 
how the terms in the model are the SAME as those from the linear mixed model in Table 10.1/ 
C. To test the cross level interaction, examine if "relfreq*grpMtheolbib" is significant:  
 
proc mixed data=mlm_complete noclprint covtest method=ml ic; 
title'H1: Model1: RI: bridge'; 
class church; 
model socprior = theolbib relfreq grpMtheolbib relfreq*grpMtheolbib / 
solution ; 
random intercept relfreq / subject = church type=un; 
title 'Model III: Cross-level interaction model'; 
run; 
 
 
Breaking down the interaction in liberal and conservative congregations 
Create a variable to group congregations as liberal or conservative.  This syntax is in 
"creating_multilevel" where a variable "grptheolbiblibcon" was created where "0" indicates a 
conservative congregation, and "1" a liberal congregation. Run the regressions between 
frequency of religious attendance and SJ prioritization separately for conservative and liberal 
congregations, to see how the associations are different in the different types of congregations: 

proc sort data=mlm_complete; 
by grptheolbiblibcon; 
run; 
 
proc reg data=mlm_complete; 
by grptheolbiblibcon; 
model socprior = relfreq; 
title 'regression for conservative (0) and liberal (1) congregations'; 
run; 
 
The warning about the variable _NAME_ or _Type_ can be ignored. 

SAS SYNTAX: MLM LOGISTIC PARTICIPATION 
 
OVERVIEW 
This syntax is for modeling the logistic outcome of SJ participation. See the "Basic_SAS" 
section of this Guide for instructions on loading data sets. Put the path to your data set in the 
“libname” statement: 
 
libname sasdata "C:\INSERT YOUR PATH TO THE FOLDER WHERE THE DATA FILES ARE 
HERE"; 
run; 
 
data mlm_complete; 
set sasdata.mlm_complete; 
run; 
 
proc sort data=mlm_complete; 
by church; 
run; 
 
 
EXPLANATION OF SAS CODE 
 
Model 
Thus in SAS we use “proc glimmix” as displayed below: 
    
 *proc glimmix data=example2  method=laplace ic=pq; 
 *class church; 
 *model socpartbin = relfreq relfreq*grpMtheolbib / solution link=logit 
dist=bin; 
 *random intercept relfreq / subject = church type=un; 
 *run; 
 
The SAS code is almost identical to the “proc mixed” used for continuous outcomes. However, 
the method of estimation is specified of “laplace” as this is a maximum likelihood estimation 
technique. Also, in the solution statement a specific “link=logit” and “dist=bin” have been 
specified to indicate that the data are logistic, and that a binomial distribution should be used. 
 
 
Null model 
For logistic outcomes it is not appropriate to calculate the intraclass correlation, and thus no null 
model is needed. 
 
THIS EXAMPLE TESTS IF FREQUENCY OF RELIGIOUS ATTENDANCE PREDICTS 
SOCIAL JUSTICE PARTICIPATION.  BRIDGING IS NOT USED SINCE IT IS NON-
SIGNIFICANT. 
  
Grand mean centered 
Grand mean centering is used to test the contextual effect of congregational religious frequency, 
or how much individual level social justice participation can be predicted by congregational 

religious frequency of attendance over and above individual level frequency of religious 
attendance. The contextual effect here is non-significant: 
  
proc glimmix data=mlm_complete  method=laplace ic=pq; 
class church; 
model socpartbin = relfreqgrand grpMrelfreq / solution link=logit dist=bin; 
random intercept / subject = church type=un; 
title 'grand-mean centered model'; 
run; 
 
Group mean centered 
Group mean centered is used to tease apart within and between effects, where “grpMrelfreq” 
represents the between effect: 
 
proc glimmix data=mlm_complete  method=laplace ic=pq; 
class church; 
model socpartbin = relfreqgroup grpMrelfreq / solution link=logit dist=bin; 
random intercept / subject = church type=un; 
title 'group-mean centered model'; 
run; 
 
CROSS LEVEL INTERACTION 
The cross level interaction tests if congregational theological orientation moderates the 
association between frequency of religious participation and social justice participation.  Note 
how the terms in the model are the SAME as those from the linear mixed model in Table 10.1/C. 
To test the cross level interaction, examine if “relfreq*grpMtheolbib” is significant:  
 
**model CROSS LEVEL INTERACTION; 
proc sort data=mlm_complete; 
by socpartbin; 
run; 
 
proc glimmix data=mlm_complete method=laplace ic=pq; 
class church; 
model socpartbin = theolbib relfreq grpMtheolbib relfreq*grpMtheolbib / 
solution link=logit dist=bin; 
random intercept relfreq / subject = church type=un; 
title 'cross-level interaction model'; 
run; 
 
 
Breaking down the interaction in liberal and conservative congregations 
Create a variable to group congregations as liberal or conservative.  This syntax is in 
“creating_multilevel” where a variable “grptheolbiblibcon” was created where “0” indicates a 
conservative congregation, and “1” a liberal congregation. Run the regressions between 
frequency of religious attendance and SJ participation separately for conservative and liberal 
congregations, to see how the associations are different in the different types of congregations: 
 
proc sort data=mlm_complete; 
by grptheolbiblibcon; 
run; 
 
proc logistic data=mlm_complete; 

by grptheolbiblibcon; 
model socpartbin(event='1') = relfreq; 
title 'Logistic regression for conservative (0) and liberal (1) 
congregations'; 
run; 
 
Follow-up logistic curve 
Getting the logistic curves can be rather tricky, so the code is given below: 
 
proc sort data=mlm_complete; 
by grptheolbiblibcon; 
run; 
 
goptions reset=(axis, legend, pattern, symbol, title, footnote) norotate 
         hpos=0 vpos=0 htext=2 ftext=swiss ctext=black target= gaccess= 
gsfmode=  
         colors= csymbol=value interpol=spline; 
 
axis1  color=black   width=2.2  label=('Frequency of Religious Participation' 
);  
axis2  color=black   width=2.2  label=(angle=90 'Social Justic Participation' 
);  
 
proc logistic data=mlm_complete order=data; 
by grptheolbiblibcon; 
model socpartbin(event='1') = relfreq; 
output out = newaa p=pred; 
run; 
 
proc sort data=newaa; 
by grptheolbiblibcon relfreq; 
 
proc gplot data=newaa; 
symbol1 interpol=spline value=triangle; 
symbol2 interpol=spline value=circle; 
plot pred*relfreq=grptheolbiblibcon / haxis=axis1 vaxis=axis2 frame; 
  title1 'Cong. Theological Orientation Moderating Frequency of Religious 
Participation and SJ Participation'; 
run; 
 
If you are unable to see the graph, go to the "view" menu in the upper left hand part of the 
screen, and select "graph". The graph should then be visible. 

